<commit_before># Creates the variables for setting up GFM auto-completion

window.GitLab ?= {}
GitLab.GfmAutoComplete ?= {}

# Emoji
data      = []
template  = "<li data-value='${insert}'>${name} <img alt='${name}' height='20' src='${image}' width='20' /></li>"
GitLab.GfmAutoComplete.Emoji = {data, template}

# Team Members
data      = []
url     = '';
params  = {private_token: '', page: 1}
GitLab.GfmAutoComplete.Members = {data, url, params}

# Add GFM auto-completion to all input fields, that accept GFM input.
GitLab.GfmAutoComplete.setup = ->
  input = $('.js-gfm-input')

  # Emoji
  input.atWho ':',
    data: GitLab.GfmAutoComplete.Emoji.data,
    tpl: GitLab.GfmAutoComplete.Emoji.template

  # Team Members
  input.atWho '@', (query, callback) ->
    (getMoreMembers = ->
      $.getJSON(GitLab.GfmAutoComplete.Members.url, GitLab.GfmAutoComplete.Members.params)
        .success (members) ->
          # pick the data we need
          newMembersData = $.map(members, (m) -> m.name )

          # add the new page of data to the rest
          $.merge(GitLab.GfmAutoComplete.Members.data, newMembersData)

          # show the pop-up with a copy of the current data
          callback(GitLab.GfmAutoComplete.Members.data[..])

          # are we past the last page?
          if newMembersData.length is 0
            # set static data and stop callbacks
            input.atWho '@',
              data: GitLab.GfmAutoComplete.Members.data
              callback: null
          else
            # get next page
            getMoreMembers()

      # so the next request gets the next page
      GitLab.GfmAutoComplete.Members.params.page += 1
    ).call()<commit_msg>fix gfm autocomplete for usernames<commit_after># Creates the variables for setting up GFM auto-completion

window.GitLab ?= {}
GitLab.GfmAutoComplete =
  # Emoji
  Emoji:
    data: []
    template: '<li data-value="${insert}">${name} <img alt="${name}" height="20" src="${image}" width="20" /></li>'

  # Team Members
  Members:
    data: []
    url: ''
    params:
      private_token: ''
      page: 1
    template: '<li data-value="${username}">${username} <small>${name}</small></li>'

  # Add GFM auto-completion to all input fields, that accept GFM input.
  setup: ->
    input = $('.js-gfm-input')

    # Emoji
    input.atWho ':',
      data: @Emoji.data
      tpl: @Emoji.template

    # Team Members
    input.atWho '@',
      tpl: @Members.template
      callback: (query, callback) =>
        (getMoreMembers = =>
          $.getJSON(@Members.url, @Members.params).done (members) =>
            # pick the data we need
            newMembersData = $.map(members, (m) ->
              username: m.username
              name: m.name
            )

            # add the new page of data to the rest
            $.merge(@Members.data, newMembersData)

            # show the pop-up with a copy of the current data
            callback(@Members.data[..])

            # are we past the last page?
            if newMembersData.length is 0
              # set static data and stop callbacks
              input.atWho '@',
                data: @Members.data
                callback: null
            else
              # get next page
              getMoreMembers()

          # so the next callback requests the next page
          @Members.params.page += 1
        ).call()
{'fraction_non_alphanumeric': 0.12652212652212652, 'fraction_numerical': 0.004158004158004158, 'mean_word_length': 2.127205199628598, 'pattern_counts': {'":': 0, '<': 13, '<?xml version=': 0, '>': 23, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '3307972', 'n_tokens_mistral': 1041, 'n_tokens_neox': 989, 'n_words': 277}